

## D03ECF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

D03ECF uses the Strongly Implicit Procedure to calculate the solution to a system of simultaneous algebraic equations of seven-point molecule form on a three-dimensional topologically-rectangular mesh. ('Topological' means that a polar grid, for example, can be used if it is equivalent to a rectangular box.)

### 2 Specification

```

SUBROUTINE D03ECF(N1, N2, N3, N1M, N2M, A, B, C, D, E, F, G, Q, T,
1          APARAM, ITMAX, ITCOUN, ITUSED, NDIR, IXN, IYN,
2          IZN, CONRES, CONCHN, RESIDS, CHNGS, WRKSP1,
3          WRKSP2, WRKSP3, WRKSP4, IFAIL)
  INTEGER   N1, N2, N3, N1M, N2M, ITMAX, ITCOUN, ITUSED,
1          NDIR, IXN, IYN, IZN, IFAIL
  real      A(N1M,N2M,N3), B(N1M,N2M,N3), C(N1M,N2M,N3),
1          D(N1M,N2M,N3), E(N1M,N2M,N3), F(N1M,N2M,N3),
2          G(N1M,N2M,N3), Q(N1M,N2M,N3), T(N1M,N2M,N3),
3          APARAM, CONRES, CONCHN, RESIDS(ITMAX),
4          CHNGS(ITMAX), WRKSP1(N1M,N2M,N3),
5          WRKSP2(N1M,N2M,N3), WRKSP3(N1M,N2M,N3),
6          WRKSP4(N1M,N2M,N3)

```

### 3 Description

Given a set of simultaneous equations

$$Mt = q \quad (1)$$

(which could be nonlinear) derived, for example, from a finite difference representation of a three-dimensional elliptic partial differential equation and its boundary conditions, the routine determines the values of the dependent variable  $t$ .  $M$  is a square  $(n_1 \times n_2 \times n_3)$  by  $(n_1 \times n_2 \times n_3)$  matrix and  $q$  is a known vector of length  $(n_1 \times n_2 \times n_3)$ .

The equations must be of seven-diagonal form:

$$a_{ijk}t_{ij,k-1} + b_{ijk}t_{i,j-1,k} + c_{ijk}t_{i-1,jk} + d_{ijk}t_{ijk} + e_{ijk}t_{i+1,jk} + f_{ijk}t_{i,j+1,k} + g_{ijk}t_{ij,k+1} = q_{ijk}$$

for  $i = 1, 2, \dots, n_1$ ;  $j = 1, 2, \dots, n_2$  and  $k = 1, 2, \dots, n_3$ , provided that  $d_{ijk} \neq 0.0$ .

Indeed, if  $d_{ijk} = 0.0$ , then the equation is assumed to be:

$$t_{ijk} = q_{ijk}.$$

The system is solved iteratively from a starting approximation  $t^{(1)}$  by the formulae:

$$r^{(n)} = q - Mt^{(n)}$$

$$Ms^{(n)} = r^{(n)}$$

$$t^{(n+1)} = t^{(n)} + s^{(n)}.$$

Thus  $r^{(n)}$  is the residual of the  $n$ th approximate solution  $t^{(n)}$ , and  $s^{(n)}$  is the up-date change vector.

The calling program supplies an initial approximation for the values of the dependent variable in the array T, the coefficients of the seven-point molecule system of equations in the arrays A, B, C, D, E, F and G, and the source terms in the array Q. The routine derives the residual of the latest approximate

solution, and then uses the approximate *LU* factorization of the Strongly Implicit Procedure with the necessary acceleration parameter adjustment by calling D03UBF at each iteration. D03ECF combines the newly derived change with the old approximation to obtain the new approximate solution for  $t$ . The new solution is checked for convergence against the user-supplied convergence criteria, and if these have not been satisfied, the iterative cycle is repeated. Convergence is based on both the maximum absolute normalised residuals (calculated with reference to the previous approximate solution as these are calculated at the commencement of each iteration) and on the maximum absolute change made to the values of  $t$ .

Problems in topologically non-rectangular-box-shaped regions can be solved using the routine by surrounding the region by a circumscribing topologically rectangular box. The equations for the nodal values external to the region of interest are set to zero (i.e.,  $d_{ijk} = t_{ijk} = 0$ ) and the boundary conditions are incorporated into the equations for the appropriate nodes.

If there is no better initial approximation when starting the iterative cycle, one can use an array of zeros as the initial approximation.

The routine can be used to solve linear elliptic equations in which case the arrays A, B, C, D, E, F, G and Q remain constant and for which a single call provides the required solution. It can also be used to solve nonlinear elliptic equations, in which case some or all of these arrays may require updating during the progress of the iterations as more accurate solutions are derived. The routine will then have to be called repeatedly in an outer iterative cycle. Dependent on the nonlinearity, some under-relaxation of the coefficients and/or source terms may be needed during their recalculation using the new estimates of the solution.

The routine can also be used to solve each step of a time-dependent parabolic equation in three space dimensions. The solution at each time step can be expressed in terms of an elliptic equation if the Crank–Nicolson or other form of implicit time integration is used.

Neither diagonal dominance, nor positive definiteness, of the matrix  $M$  formed from the arrays A, B, C, D, E, F, G is necessary to ensure convergence.

For problems in which the solution is not unique in the sense that an arbitrary constant can be added to the solution (for example Poisson's equation with all Neumann boundary conditions), a parameter is incorporated so that the solution can be rescaled. A specified nodal value is subtracted from the whole solution  $t$  after the completion of every iteration. This keeps rounding errors to a minimum for those cases when convergence is slow. For such problems there is generally an associated compatibility condition. For the example mentioned this compatibility condition equates the total net source within the region (i.e., the source integrated over the region) with the total net outflow across the boundaries defined by the Neumann conditions (i.e., the normal derivative integrated along the whole boundary). It is very important that the algebraic equations derived to model such a problem implement accurately the compatibility condition. If they do not, a net source or sink is very likely to be represented by the set of algebraic equations and no steady-state solution of the equations exists.

## 4 References

- [1] Jacobs D A H (1972) The strongly implicit procedure for the numerical solution of parabolic and elliptic partial differential equations *Note RD/L/N66/72* Central Electricity Research Laboratory
- [2] Stone H L (1968) Iterative solution of implicit approximations of multi-dimensional partial differential equations *SIAM J. Numer. Anal.* **5** 530–558
- [3] Weinstein H G, Stone H L and Kwan T V (1969) Iterative procedure for solution of systems of parabolic and elliptic equations in three dimensions *Industrial and Engineering Chemistry Fundamentals* **8** 281–287

## 5 Parameters

- 1: N1 — INTEGER *Input*  
*On entry:* the number of nodes in the first co-ordinate direction,  $n_1$ .  
*Constraint:* N1 > 1.

- 2:** N2 — INTEGER *Input*  
*On entry:* the number of nodes in the second co-ordinate direction,  $n_2$ .  
*Constraint:*  $N2 > 1$ .
- 3:** N3 — INTEGER *Input*  
*On entry:* the number of nodes in the third co-ordinate direction,  $n_3$ .  
*Constraint:*  $N3 > 1$ .
- 4:** N1M — INTEGER *Input*  
*On entry:* the first dimension of all the three-dimensional arrays, as declared in the (sub)program from which D03ECF is called.  
*Constraint:*  $N1M \geq N1$ .
- 5:** N2M — INTEGER *Input*  
*On entry:* the second dimension of all the three-dimensional arrays, as declared in the (sub)program from which D03ECF is called.  
*Constraint:*  $N2M \geq N2$ .
- 6:** A(N1M,N2M,N3) — *real* array *Input*  
*On entry:* A( $i, j, k$ ) must contain the coefficient of  $t_{ij,k-1}$  in the ( $i, j, k$ )th equation of the system (1), for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ . The elements of A for  $k = 1$  must be zero after incorporating the boundary conditions, since they involve nodal values from outside the box.
- 7:** B(N1M,N2M,N3) — *real* array *Input*  
*On entry:* B( $i, j, k$ ) must contain the coefficient of  $t_{i,j-1,k}$  in the ( $i, j, k$ )th equation of the system (1), for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ . The elements of B for  $j = 1$  must be zero after incorporating the boundary conditions, since they involve nodal values from outside the box.
- 8:** C(N1M,N2M,N3) — *real* array *Input*  
*On entry:* C( $i, j, k$ ) must contain the coefficient of  $t_{i-1,j,k}$  in the ( $i, j, k$ )th equation of the system (1), for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ . The elements of C for  $i = 1$  must be zero after incorporating the boundary conditions, since they involve nodal values from outside the box.
- 9:** D(N1M,N2M,N3) — *real* array *Input*  
*On entry:* D( $i, j, k$ ) must contain the coefficient of  $t_{ijk}$  (the ‘central’ term) in the ( $i, j, k$ )th equation of the system (1), for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ . The elements of D are checked to ensure that they are non-zero. If any element is found to be zero, the corresponding algebraic equation is assumed to be  $t_{ijk} = q_{ijk}$ . This feature can be used to define the equations for nodes at which, for example, Dirichlet boundary conditions are applied, or for nodes external to the problem of interest. Setting D( $i, j, k$ ) = 0.0 at appropriate points, and the corresponding value of Q( $i, j, k$ ) to the appropriate value, namely the prescribed value of T( $i, j, k$ ) in the Dirichlet case, or to zero at an external point.
- 10:** E(N1M,N2M,N3) — *real* array *Input*  
*On entry:* E( $i, j, k$ ) must contain the coefficient of  $t_{i+1,j,k}$  in the ( $i, j, k$ )th equation of the system (1), for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ . The elements of E for  $i = N1$  must be zero after incorporating the boundary conditions, since they involve nodal values from outside the box.

- 11:** F(N1M,N2M,N3) — *real* array *Input*  
*On entry:* F( $i, j, k$ ) must contain the coefficient of  $t_{i,j+1,k}$  in the ( $i, j, k$ )th equation of the system (1), for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ . The elements of F for  $j = N2$  must be zero after incorporating the boundary conditions, since they involve nodal values from outside the box.
- 12:** G(N1M,N2M,N3) — *real* array *Input*  
*On entry:* G( $i, j, k$ ) must contain the coefficient of  $t_{i,j,k+1}$  in the ( $i, j, k$ )th equation of the system (1), for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ . The elements of G for  $k = N3$  must be zero after incorporating the boundary conditions, since they involve nodal values from outside the box.
- 13:** Q(N1M,N2M,N3) — *real* array *Input*  
*On entry:* Q( $i, j, k$ ) must contain  $q_{ijk}$  for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ , i.e., the source-term values at the nodal points of the system (1).
- 14:** T(N1M,N2M,N3) — *real* array *Input/Output*  
*On entry:* T( $i, j, k$ ) must contain the element  $t_{ijk}$  of an approximate solution to the equations for  $i = 1, 2, \dots, N1$ ;  $j = 1, 2, \dots, N2$  and  $k = 1, 2, \dots, N3$ .  
 If no better approximation is known, an array of zeros can be used.  
*On exit:* the solution derived by the routine.
- 15:** APARAM — *real* *Input*  
*On entry:* the iteration acceleration factor. A value of 1.0 is adequate for most typical problems. However, if convergence is slow, the value can be reduced, typically to 0.2 or 0.1. If divergence is obtained, the value can be increased, typically to 2.0, 5.0 or 10.0.  
*Constraint:*  $0.0 < \text{APARAM} \leq ((N-1)^2 + (N2-1)^2 + (N3-1)^2)/3.0$ .
- 16:** ITMAX — INTEGER *Input*  
*On entry:* the maximum number of iterations to be used by the routine in seeking the solution. A reasonable value might be 20 for a problem with 3000 nodes and convergence criteria of about  $10^{-3}$  of the original residual and change.
- 17:** ITCOUN — INTEGER *Input/Output*  
*On entry:* on the first call of D03ECF, ITCOUN must be set to 0. On subsequent entries, its value must be unchanged from the previous call.  
*On exit:* its value is increased by the number of iterations used on this call (namely ITUSED). It therefore stores the accumulated number of iterations actually used.  
 For subsequent calls for the same problem, i.e., with the same N1, N2 and N3 but possibly different coefficients and/or source terms, as occur with nonlinear systems or with time-dependent systems, ITCOUN should not be reset, i.e., it must contain the accumulated number of iterations. In this way a suitable cycling of the sequence of iteration parameters is obtained in the calls to D03UBF.
- 18:** ITUSED — INTEGER *Output*  
*On exit:* the number of iterations actually used on that call.
- 19:** NDIR — INTEGER *Input*  
*On entry:* indicates whether or not the system of equations has a unique solution. For systems which have a unique solution, NDIR must be set to any non-zero value. For systems derived from problems to which an arbitrary constant can be added to the solution, for example Poisson's equation with all Neumann boundary conditions, NDIR should be set to 0 and the values of the next three parameters must be specified. For such problems the routine subtracts the value of the function derived at the node (IXN, IYN, IZN) from the whole solution after each iteration to reduce the possibility of large rounding errors. The user must also ensure for such problems that the appropriate compatibility condition on the source terms Q is satisfied. See the comments at the end of Section 3.

- 20: IXN — INTEGER** *Input*  
*On entry:* IXN is ignored unless NDIR is equal to zero, in which case it must specify the first index of the nodal point at which the solution is to be set to zero. The node should not correspond to a corner node, or to a node external to the region of interest.
- 21: IYN — INTEGER** *Input*  
*On entry:* IYN is ignored unless NDIR is equal to zero, in which case it must specify the second index of the nodal point at which the solution is to be set to zero. The node should not correspond to a corner node, or to a node external to the region of interest.
- 22: IZN — INTEGER** *Input*  
*On entry:* IZN is ignored unless NDIR is equal to zero, in which case it must specify the third index of the nodal point at which the solution is to be set to zero. The node should not correspond to a corner node, or to a node external to the region of interest.
- 23: CONRES — *real*** *Input*  
*On entry:* the convergence criterion to be used on the maximum absolute value of the normalised residual vector components. The latter is defined as the residual of the algebraic equation divided by the central coefficient when the latter is not equal to 0.0, and defined as the residual when the central coefficient is zero.  
 CONRES should not be less than a reasonable multiple of the *machine precision*.
- 24: CONCHN — *real*** *Input*  
*On entry:* the convergence criterion to be used on the maximum absolute value of the change made at each iteration to the elements of the array T, namely the dependent variable. CONCHN should not be less than a reasonable multiple of the machine accuracy multiplied by the maximum value of T attained.  
 Convergence is achieved when both the convergence criteria are satisfied. The user can therefore set convergence on either the residual or on the change, or (as is recommended) on a requirement that both are below prescribed limits.
- 25: RESIDS(ITMAX) — *real* array** *Output*  
*On exit:* the maximum absolute value of the residuals calculated at the  $i$ th iteration, for  $i = 1, 2, \dots, \text{ITUSED}$ . If the residual of the solution is sought the user must calculate this in the (sub)program from which D03ECF is called. The sequence of values RESIDS indicates the rate of convergence.
- 26: CHNGS(ITMAX) — *real* array** *Output*  
*On exit:* the maximum absolute value of the changes made to the components of the dependent variable T at the  $i$ th iteration, for  $i = 1, 2, \dots, \text{ITUSED}$ . The sequence of values CHNGS indicates the rate of convergence.
- 27: WRKSP1(N1M,N2M,N3) — *real* array** *Workspace*
- 28: WRKSP2(N1M,N2M,N3) — *real* array** *Workspace*
- 29: WRKSP3(N1M,N2M,N3) — *real* array** *Workspace*
- 30: WRKSP4(N1M,N2M,N3) — *real* array** *Workspace*
- 31: IFAIL — INTEGER** *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error or gives a warning (see Section 6).  
**For this routine**, because the values of output parameters may be useful even if IFAIL  $\neq$  0 on exit, users are recommended to set IFAIL to -1 before entry. **It is then essential to test the value of IFAIL on exit.** To suppress the output of an error message when soft failure occurs, set IFAIL to 1.

## 6 Error Indicators and Warnings

Errors or warnings specified by the routine:

IFAIL = 1

On entry, N1 < 2,  
or N2 < 2,  
or N3 < 2.

IFAIL = 2

On entry, N1M < N1,  
or N2M < N2.

IFAIL = 3

On entry, APARAM ≤ 0.0.

IFAIL = 4

On entry, APARAM > ((N1 - 1)<sup>2</sup> + (N2 - 1)<sup>2</sup> + (N3 - 1)<sup>2</sup>)/3.0.

IFAIL = 5

Convergence was not achieved after ITMAX iterations.

## 7 Accuracy

The improvement in accuracy for each iteration depends on the size of the system and on the condition of the up-date matrix characterised by the seven-diagonal coefficient arrays. The ultimate accuracy obtainable depends on the above factors and on the *machine precision*. The rate of convergence obtained with the Strongly Implicit Procedure is not always smooth because of the cyclic use of nine acceleration parameters. The convergence may become slow with very large problems. The final accuracy obtained may be judged approximately from the rate of convergence determined from the sequence of values returned in the arrays RESIDS and CHNGS and the magnitude of the maximum absolute value of the change vector on the last iteration stored in CHNGS(ITUSED).

## 8 Further Comments

The time taken by the routine per iteration is approximately proportional to N1 × N2 × N3.

Convergence may not always be obtained when the problem is very large and/or the coefficients of the equations have widely disparate values. The latter case is often associated with a near ill-conditioned matrix.

## 9 Example

To solve Laplace's equation in a rectangular box with a non-uniform grid spacing in the  $x$ ,  $y$ , and  $z$  co-ordinate directions and with Dirichlet boundary conditions specifying the function on the surfaces of the box equal to

$$e^{(1.0+x)/y(n_2)} \times \cos(\sqrt{2}y/y(n_2)) \times e^{(-1.0-z)/y(n_2)}.$$

Note that this is the same problem as that solved in the example for D03UBF. The differences in the maximum residuals obtained at each iteration between the two test runs are explained by the fact that in D03ECF the residual at each node is normalised by dividing by the central coefficient, whereas this normalisation has not been used in the example program for D03UBF.

## 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      D03ECF Example Program Text
*      Mark 19 Revised. NAG Copyright 1999.
*      .. Parameters ..
      INTEGER          N1, N2, N3, N1M, N2M, ITMAX
      PARAMETER        (N1=4,N2=5,N3=6,N1M=N1,N2M=N2,ITMAX=18)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*      .. Local Scalars ..
      real             APARAM, CONCHN, CONRES, ROOT2
      INTEGER          I, IFAIL, ITCOUN, ITUSED, IXN, IYN, IZN, J, K,
+                     NDIR
*      .. Local Arrays ..
      real             A(N1M,N2M,N3), B(N1M,N2M,N3), C(N1M,N2M,N3),
+                     CHNGS(ITMAX), D(N1M,N2M,N3), E(N1M,N2M,N3),
+                     F(N1M,N2M,N3), G(N1M,N2M,N3), Q(N1M,N2M,N3),
+                     RESIDS(18), T(N1M,N2M,N3), WRKSP1(N1M,N2M,N3),
+                     WRKSP2(N1M,N2M,N3), WRKSP3(N1M,N2M,N3),
+                     WRKSP4(N1M,N2M,N3), X(N1), Y(N2), Z(N3)
*      .. External Subroutines ..
      EXTERNAL          D03ECF
*      .. Intrinsic Functions ..
      INTRINSIC         COS, EXP, SQRT
*      .. Data statements ..
      DATA             X(1), X(2), X(3), X(4)/0.0e0, 1.0e0, 3.0e0,
+                     6.0e0/
      DATA             Y(1), Y(2), Y(3), Y(4), Y(5)/0.0e0, 1.0e0, 3.0e0,
+                     6.0e0, 10.0e0/
      DATA             Z(1), Z(2), Z(3), Z(4), Z(5), Z(6)/0.0e0, 1.0e0,
+                     3.0e0, 6.0e0, 10.0e0, 15.0e0/
*      .. Executable Statements ..
      WRITE (NOUT,*) 'D03ECF Example Program Results'
      WRITE (NOUT,*)
      ROOT2 = SQRT(2.0e0)
      APARAM = 1.0e0
      ITCOUN = 0
      NDIR = 1
      CONRES = 0.1e-5
      CONCHN = 0.1e-5
*      Set up difference equation coefficients, source terms and
*      initial approximation.
      DO 80 K = 1, N3
        DO 60 J = 1, N2
          DO 40 I = 1, N1
            IF ((I.NE.1) .AND. (I.NE.N1) .AND. (J.NE.1)
+             .AND. (J.NE.N2) .AND. (K.NE.1) .AND. (K.NE.N3)) THEN
*              Specification for internal nodes
              A(I,J,K) = 2.0e0/((Z(K)-Z(K-1))*(Z(K+1)-Z(K-1)))
              G(I,J,K) = 2.0e0/((Z(K+1)-Z(K))*(Z(K+1)-Z(K-1)))
              B(I,J,K) = 2.0e0/((Y(J)-Y(J-1))*(Y(J+1)-Y(J-1)))
              F(I,J,K) = 2.0e0/((Y(J+1)-Y(J))*(Y(J+1)-Y(J-1)))
              C(I,J,K) = 2.0e0/((X(I)-X(I-1))*(X(I+1)-X(I-1)))
              E(I,J,K) = 2.0e0/((X(I+1)-X(I))*(X(I+1)-X(I-1)))
              D(I,J,K) = -A(I,J,K) - B(I,J,K) - C(I,J,K) - E(I,J,K)
+
              - F(I,J,K) - G(I,J,K)

```

```

      Q(I,J,K) = 0.0e0
      T(I,J,K) = 0.0e0
    ELSE
      Specification for boundary nodes
*
      20      A(I,J,K) = 0.0e0
            B(I,J,K) = 0.0e0
            C(I,J,K) = 0.0e0
            E(I,J,K) = 0.0e0
            F(I,J,K) = 0.0e0
            G(I,J,K) = 0.0e0
            D(I,J,K) = 0.0e0
            Q(I,J,K) = EXP((X(I)+1.0e0)/Y(N2))*COS(ROOT2*Y(J)
+              /Y(N2))*EXP((-Z(K)-1.0e0)/Y(N2))
            T(I,J,K) = 0.0e0
      END IF
    40      CONTINUE
    60      CONTINUE
    80      CONTINUE
      WRITE (NOUT,*) 'Iteration      Maximum      Maximum'
      WRITE (NOUT,*) ' number      residual      change'
      IFAIL = 0
*
      CALL D03ECF(N1,N2,N3,N1M,N2M,A,B,C,D,E,F,G,Q,T,APARAM,ITMAX,
+        ITCOUN,ITUSED,NDIR,IXN,IYN,IZN,CONRES,CONCHN,RESIDS,
+        CHNGS,WRKSP1,WRKSP2,WRKSP3,WRKSP4,IFAIL)
*
      IF (ITUSED.NE.0) WRITE (NOUT,99999) (I,RESIDS(I),CHNGS(I),I=1,
+      ITUSED)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Table of calculated function values'
      WRITE (NOUT,*)
      WRITE (NOUT,*)
+      'K J (I T ) (I T ) (I T ) (I T )'
      WRITE (NOUT,99998) ((K,J,(I,T(I,J,K),I=1,N1),J=1,N2),K=1,N3)
      STOP
*
    99999 FORMAT (2X,I3,9X,e11.4,4X,e11.4)
    99998 FORMAT ((1X,I1,I3,1X,4(1X,I3,2X,F8.3)))
      END

```

## 9.2 Program Data

None.

## 9.3 Program Results

D03ECF Example Program Results

Iteration number	Maximum residual	Maximum change
1	0.1822E+01	0.1822E+01
2	0.9025E-02	0.1970E-01
3	0.1358E-02	0.1496E-02
4	0.4013E-04	0.3848E-04
5	0.5321E-05	0.5481E-05
6	0.2695E-06	0.2333E-06



Table of calculated function values

K	J	(I	T	)	(I	T	)	(I	T	)	(I	T	)
1	1	1	1.000	2	1.105	3	1.350	4	1.822				
1	2	1	0.990	2	1.094	3	1.336	4	1.804				
1	3	1	0.911	2	1.007	3	1.230	4	1.661				
1	4	1	0.661	2	0.731	3	0.892	4	1.205				
1	5	1	0.156	2	0.172	3	0.211	4	0.284				
2	1	1	0.905	2	1.000	3	1.221	4	1.649				
2	2	1	0.896	2	0.990	3	1.210	4	1.632				
2	3	1	0.825	2	0.912	3	1.114	4	1.503				
2	4	1	0.598	2	0.662	3	0.809	4	1.090				
2	5	1	0.141	2	0.156	3	0.190	4	0.257				
3	1	1	0.741	2	0.819	3	1.000	4	1.350				
3	2	1	0.733	2	0.811	3	0.991	4	1.336				
3	3	1	0.675	2	0.747	3	0.913	4	1.230				
3	4	1	0.490	2	0.543	3	0.664	4	0.892				
3	5	1	0.116	2	0.128	3	0.156	4	0.211				
4	1	1	0.549	2	0.607	3	0.741	4	1.000				
4	2	1	0.543	2	0.601	3	0.734	4	0.990				
4	3	1	0.500	2	0.554	3	0.677	4	0.911				
4	4	1	0.363	2	0.402	3	0.492	4	0.661				
4	5	1	0.086	2	0.095	3	0.116	4	0.156				
5	1	1	0.368	2	0.407	3	0.497	4	0.670				
5	2	1	0.364	2	0.403	3	0.492	4	0.664				
5	3	1	0.335	2	0.371	3	0.454	4	0.611				
5	4	1	0.243	2	0.270	3	0.330	4	0.443				
5	5	1	0.057	2	0.063	3	0.077	4	0.105				
6	1	1	0.223	2	0.247	3	0.301	4	0.407				
6	2	1	0.221	2	0.244	3	0.298	4	0.403				
6	3	1	0.203	2	0.225	3	0.274	4	0.371				
6	4	1	0.148	2	0.163	3	0.199	4	0.269				
6	5	1	0.035	2	0.038	3	0.047	4	0.063				

---